Basic fur das fischertechnik Robo Interface (Version 0.9.4 Beta) 31.07.2007

Vorwort

Diese Software entstand aus der Idee heraus, eine Scriptsprache fiir das Robo Interface zu entwickeln, die
auch downloadfahig ist — d.h.: die auch autonom ohne PC im Interface funktioniert. Zwar existiert bereits eine
downloadfahige Scriptsprache in Form von C, allerdings ist dies sehr fehleranfallig und fiir Anfanger ganzlich
ungeeignet.

Es ist in Zukunft geplant, die Sourcen zu verdéffentlichen — allerdings miissen die erst einmal bereinigt und
lesbar gemacht werden ;-)

Ich kann leider nicht garantieren, dass die vorliegende Software véllig fehlerfrei ist. Eine Benutzung geschieht
deshalb auf eigenes Risiko, eine Haftung bei Folgeschaden durch die Benutzung dieser Software schlielRe
ich aus.

Momentan hat die Software Beta-Status.

Bitte diese Anleitung komplett durchlesen!
Sollten Teile dieser Beschreibung nicht verstandlich sein, bitte ich um Nachricht.
Historie (Anderungen gegeniiber der Vorgingerversion)

— Befehl ,edge® ist nun eine Funktion mit zusatzlichen Parametern

— Ausgaben mit dem Print-Befehl werden nun direkt auf der seriellen Schnittstelle ausgegeben
(ohne das Overhead der Firmware-Message-Funktion)

— Hinzugekommen sind put, get() fur serielle Kommunikation, len() zur Ermittlung der Lange eines Strings

— Die Ein- und Ausgabefunktionen fur die digitalen Ein- und Ausgénge wurden von input() und output auf in
() und out geandert, da input() nun fir die serielle Eingabe benutzt wird.

— Die Endung der Basic-Dateien wurde von bas auf rib (Robo Interface Basic) geandert, damit die Endung
nicht mit anderer Software kollidiert.

— Baud-Befehl ist hinzugekommen

— Der Bytecode kann nun entweder in Flash2 oder in den RAM geladen werden. Der Interpreter verbleibt
immer im Flash1 und muss nicht jedesmal mitgeladen werden.

1. Wie schreibe und compiliere ich Programme?

Zur Installation des Basic-Pakets die ZIP-Datei einfach in ein Laufwerk entpacken. Es wird ein Verzeichnis
-Robolnt_Basic_Compiler erstellt.

AnschlieBend die Datei anpassen.exe ausfihren. Hierbei werden nur die Verzeichnisse beim ,Programmers
Notepad® angemeldet, es wird nichts in der Registrierung des Computers geandert.

Wird anpassen.exe nicht ausgefiihrt, sind die Dateipfade nicht richtig definiert und der Aufruf des
Compilers aus dem Notepad heraus funktioniert nicht.

Bei Nutzung eines eigenen Editors ist die Datei anpassen.exe nicht notwendig.

Im Paket ist der kostenlose Editor "Programmers Notepad" enthalten, der auch unter
http://www.pnotepad.org heruntergeladen werden kann. Er bietet Syntax-Highlighting und Tools fur den
einfachen Aufruf der bendtigten Programme.

Das gesamte Programmpaket bendtigt nur User-Rechte zur Installation und kann vollstandig entfernt werden,
indem der von Programmers Notepad angelegte Ordner ,Echo Software” und der Ordner
.Robolnt_Basic_Compiler* geldéscht werden.

An der Dateistruktur des Verzeichnisses "Robolnt_Basic_Compiler" sollte nichts verandert werden, da der
Editor diese Struktur erwartet und nur dann richtig funktioniert.

1.1 Schreiben eines Programms

Starten des Editors mit Doppelklick auf "pn.exe" im Unterverzeichnis ,pn“. Beim ersten Aufruf speichert der
Editor seine Konfigurationsdaten in "C:\Dokumente und Einstellungen\(Username)\Anwendungsdaten\Echo
Software". Hierbei werden auch die Tools automatisch definiert.

Nun kann das Programm eingegeben oder eines der Beispielprogramme im Ordner "Testprogramme"
geladen werden. Wichtig: Das Label main: muss auf jeden Fall im Programm vorhanden sein, da der
Compiler hierduch mitgeteilt bekommt, wo das Hauptprogramm beginnt.



Das Programm muss mit der Endung .rib abgespeichert werden. Der Compiler erwartet das
Programm als Datei, das Programm muss also unbedingt vor dem Compilieren gespeichert werden.
Wurde das Programm schon einmal gespeichert und wird nur verandert, ist das Programmers Notepad so
voreingestellt, dass es die veranderte Datei bei Druck auf ,F5“ automatisch abspeichert.

.Programmers Notepad* ist bereits so installiert, dass der Compiler mit Funktionstasten aufgerufen werden
kann, um den schnellen Beginn mit der Basic-Programmierung zu ermdglichen, auerdem werden zur
besseren Lesbarkeit die Befehle fett angezeigt (,Syntax Highlighting®).

Falls man mit Programmers Notepad nicht arbeiten mochte, eignet sich jeder beliebige Texteditor zur
Programmeingabe. Hinweise zum Aufruf des Compilers am Ende dieses Kapitels.

1.2 Compilieren eines Programms

Mit Klick auf "Tools" und dann auf "[Robolnt] compile" wird das Programm compiliert und gelinkt. Beim Link-
Vorgang wird der Interpreter automatisch zum generierten Bytecode hinzugefiigt, so dass immer ein
lauffahiges Programm entsteht. Dies ist auch der Grund, weshalb die Programme mehr als 32 KByte haben.
Diese Datei hat die Endung ,..hex“.

Im Output-Fenster am unteren Rand wird entweder die erfolgreiche Compilierung oder ein Fehler angezeigt.
Bei der Ausgabe "Process Exit Code: 0" ist alles in Ordnung. Der Compiler kann auch mit der Taste "F5"
aufgerufen werden. Zusatzliche Optionen (nur fur "Fortgeschrittene"):

- compile (Line Numbers) : Fugt Informationen der Zeilennummern hinzu. Bei Runtime-Fehlern wird diese
Information benutzt, um die Zeilennummer auszugeben, in der der Fehler passierte.

- compile (Detailed) : Der Compiler gibt alle Zwischenschritte aus, die er beim Ubersetzen des Codes
bendtigt. Dient zum Debuggen des Compilers.

Beim "Process Exit Code: 1" ist ein Fehler beim Compilieren aufgetreten, tblicherweise sollte eine
entsprechende Fehlermeldung angezeigt werden, die die Ursache des Fehlers zeigt.

1.3 Ubertragen eines Programms zum Interface

Wahrend des compilierens wird ein "hex"-File erzeugt, das den Bytecode enthalt. Das Hex-File hat den
gleichen Namen, wie der Quelltext.

Durch Aufruf des Tools "Uploader" wird der Bytecode in das Robolnterface hochgeladen. Um so wenige
Parameter wie mdglich festlegen zu missen, sucht der Uploader nach dem ersten Robolnterface an der
USB-Schnittstelle und benutzt dieses. Sind mehrere Interfaces angeschlossen, kommt es darauf an, welches
von Windows als ,erstes” Interface in der USB-Reihenfolge definiert wird. Soll bei mehreren Robolnterface
ein bestimmtes angesprochen werden, so missen die anderen abgeschaltet werden.

Der Bytecode kann entweder in das RAM oder in den Flash-Speicher 2 geladen werden. Wahrend des
Upload wird automatisch berpruft, ob sich im Flash-Speicher 1 der Interpreter befindet. Ist dies nicht der
Fall, wird er automatisch nachgeladen.

Der RAM-Speicher ist fir Versuche und wahrend der Entwicklung eines Programms besser geeignet, da der
Flash-Speicher eine begrenzte Anzahl Schreibzyklen hat.

Der Interpreter ist so eingestellt, dass er zuerst priift, ob sich ein Programm im RAM befindet. Ist dort eines
vorhanden, wird es ausgefuhrt, auch dann, wenn in Flash-Speicher 2 ein Programm ist.

1.4 Ausfiihren des Programms

Das Programm ,Control“ verbindet sich mit dem ersten Robolnterface, das auf der USB-Schnittstelle
gefunden wird.

Achtung: Die Firmware 16scht den RAM, wenn ein Programm im Flash ausgefiihrt wird.

Programme im RAM haben immer Vorrang, auch wenn Flash2 gestartet wird, iiberpriift der Interpreter
zuerst den RAM und fiihrt ein dort gespeichertes Programm aus.

Das Control-Programm 6ffnet sich bei Verwendung des Uploaders automatisch.

Momentan noch nicht moéglich, aber geplant, ist der automatische Programmestart fir Programme in Flash2.

1.5 Anzeige von Meldungen des Interface

Der Basic-Interpreter gibt Status- und Fehlermeldungen Uber die serielle Schnittstelle des Robo-Interface
aus. Diese Funktion kann genutzt werden, indem das mitgelieferte Kabel an den PC angeschlossen wird. Mit
einem beliebigen Terminal-Programm (eingestellt auf 38400 Baud, 8 Bit, 1 Stopbit, keine Parity) kdnnen dann
die Ausgaben auf dem PC angezeigt werden. Ebenfalls werden Daten, die mit dem "print" Befehl
ausgegeben werden, angezeigt, auch Runtime-Fehler werden (ber die serielle Schnittstelle ausgegeben.
Beim normalen Programmende blinkt die rote Fehler-LED des Interface 1x. Sollte ein Runtime-Fehler
auftauchen,schaltet die rote LED auf dauerblinken.



1.6 Aufruf des Compilers

Fallls man den Compiler von Hand oder aus einem anderen Editor heraus benutzen méchte, ist dies ohne
weiteres durch den direkten Aufruf mdglich.

Der Compiler erwartet bei Aufruf insgesamt 3 Parameter:

compiler [dateiname] [Zeilennummern] [Debug-Option]

[dateiname] Dateiname des Quelltextes, die Endung .rib darf nicht angegeben werden, sie wird
automatisch erganzt.
[Zeilennummern] 0 oder 1. Bei 1 werden Zeilennummern in den Bytecode einkodiert, bei einem

Runtime-Fehler wird dann die Zeilennummer des Source angegeben, in der der
Fehler vorkommt. Der Bytecode wird dadurch allerdings groRer.

{Debug-Option] 0, 1 oder 2. Bei 0 werden nur die notigsten Ausgaben wahrend des Compilierens
ausgegeben, bei 1 etwas mehr und bei 2 sehr ausfihrlich. Die Optionen 1 und 2 sind
eigentlich nur zum Debuggen des Compilers gedacht, aber wer mdchte, kann sich
damit genau anschauen, wie der Compiler den Quellcode analysiert.

Beispiel:
compiler testprogramm 0 0
1.7 Alternativer Upload

Alternativ kann der Upload auch mit dem von der Fa. Knobloch bereitgestellten FtLoader erfolgen. Hierbei
muss die Datei ,interpreter.rom” geladen werden (ist automatisch fur Flash1 adressiert). AnschlieRend kann
der Bytecode als .hex nachgeladen werden, er ist fir das RAM adressiert.

2. Funktionsprinzip

Das Basic besteht aus zwei Komponenten:
— Einem Compiler, der den Quelltext in einen Bytecode umsetzt
— Einem Interpreter, der den Bytecode ausfihrt.

Bei der Planung des Basic habe ich viele verschiedene Mdglichkeiten gepriift und — meines Erachtens — die
bestmdgliche verwirklicht. In der Anfangsphase sollte es ein reinrassiger Interpreter werden, dies hatte den
Vorteil gehabt, dass man theoretisch ohne einen PC ausgekommen ware (liber die serielle Schnittstelle ware
der Anschluss eines Displays und einer Tastatur durchaus mdglich). Allerdings hatte dies bedeutet, dass der
Prozessor des Interface sehr stark mit den Ubersetzungsaufgaben beschéftigt gewesen ware. Eine
Zwischenldsung ist die Umsetzung von Befehlen in Tokens, d.h. der Klartext der Befehle wird in Bytecode
umgesetzt (dies war z.B. das Prinzip des C64-Basic). Bei dieser Methode bleiben aber immer noch:
Berechnung der Sprungadressen, Berechnung der Variablenadressen, Syntaxprifung etc. Gerade die
Syntaxprifung ist ein Punkt, der fiir einen vom Interpreter getrennten Compiler sprach.

In dem nun vorliegenden Programm sind im Bytecode alle Adressen von Variablen, Funktionen und
sonstigen Springen vorberechnet und auch die Konstanten liegen nicht als Ascii, sondern im Binarformat
vor, was die Ausflhrungszeiten positiv beeinflusst.

Theoretisch ware es auf Grundlage des Bytecodes auch mdglich, andere Sprachen (z.B. Pascal 0.4.) zu
realisieren.

Beim Ubersetzungsvorgang des Compilers wird ein .hex-File erzeugt, das mit dem von ft gelieferten
.FtLoader” in das Robo Interface libertragen werden kann. Das .hex-File beinhaltet den Interpreter und den
Bytecode, ist also ohne weitere Downloads funktionstlichtig.

2.1 Features

In den letzten Monaten hat sich die Liste der Features von Woche zu Woche geandert — meistens wurde sie
verlangert ;-). Mittlerweile bin ich der Uberzeugung, dass man mit diesem Basic durchaus ein interessantes
Werkzeug zur Verfiigung hat. Bei der Realisierung bin ich hauptsachlich von meinen Bedurfnissen
ausgegangen — falls jemand ein Feature vemisst, das realisierbar scheint, kann ich nochmal etwas
nachlegen.

Eine kurze Liste der realisierten Features (ausfihrliche Information gibt es in der Befehlsreferenz)

— Multitasking mit bis zu 10 Tasks
— lokale Variablen innerhalb von Tasks und Funktionen



— Funktionen mit Parametern (auch Rickgabe)

— Funktionen rekursiv aufrufbar

— Funktionen mehrfach aus verschiedenen Tasks aufrufbar (reentrant)

— Ausgabe von Zahlen und Texten Uber die serielle Schnittstelle zum Debuggen
— Arrays mit bis zu zwei Dimensionen (nur global méglich)

— Datentypen Byte, Integer, Long, Float, Double und String

— Trigonometrische Funktionen mit Double-Genauigkeit

— Unterstutzung der Hardware des Robo Interface mit entsprechenden Befehlen

2.2 Multitasking

Das mit Sicherheit interessanteste Feature ist das Multitasking. Nachdem ich es in C aufgegeben habe, weil
es sehr viel Durcheinander auf dem Stack gegeben hat, war die Realisierung innerhalb eines Interpreters
geradezu einfach. Die Tasks (bis zu 10 + Hauptprogramm) werden scheinbar gleichzeitig ausgefiihrt und
auch Wartebefehle haben darauf keinen Einfluss.

Die einzige Einschrankung: Fur einen Taskwechsel muss der mathematische Stack leer sein. Vor allem bei
rekursivem Aufruf von Funktionen kommt es vor, dass ein Berechnungsergebnis auf dem mathematischen
Stack verbleibt, dies hat zur Folge, dass die komplette Rekursion erst aufgeldst werden muss, bevor der
nachste Task ausgefiihrt werden kann. Diese Eigenheit Iasst sich ohne weiteres durch eine Erweiterung
I6sen, allerdings sehe ich es eher als Feature, weil man hierdurch eine Art Priorisierung erreichen kann.

2.3 Globale und lokale Variablen

Neben den globalen Variablen, die im gesamten Programm sichtbar sind, kdnnen fiir jede Funktion und
jeden Task separat lokale Variablen definiert werden. Die lokalen Variablen sind nur innerhalb des Tasks /
der Funktion sichtbar, in dem / der sie definiert wurden, nicht dariiber hinaus.

2.4 Funktionen mit Parametern

Funktionen kdnnen mit oder ohne Parameter definiert werden. Werden Parameter definiert, sind diese
automatisch als lokale Variable innerhalb der Funktion definiert. Ein Riickgabeparameter wird mit dem return
Befehl tibergeben.

Dadurch, dass lokale Variablen fir jeden Task getrennt gespeichert werden, kénnen Funktionen aus
mehreren Task unabhangig voneinander und gleichzeitig aufgerufen werden. Die Aufrufe beeinflussen sich
nicht gegenseitig. In diesem Zusammenhang ist aber Vorsicht bei globalen Variablen geboten.

Die restlichen Features sind im Kapitel 2 besprochen.

2.5 Bekannte Bugs

Wo Licht ist, ist auch Schatten — es gibt noch einige Bugs, die mir zwar bekannt sind, aber bei denen
momentan einfach die Zeit fehlt.

Eine aktuelle Liste ist im Text-File ,bekannte_bugs.txt“ abgelegt. Diese Liste sollte unbedingt gelesen
werden, bevor man mit dem Programmieren anfangt. Ansonsten bin ich nattrlich immer froh, wenn mir Bugs
mitgeteilt werden.

3. Programmieren in BASIC

Ich gehe davon aus, dass die grundlegenden BASIC-Begriffe und Programmiertechniken bekannt sind.

Die wichtigsten Unterschiede zu einem Standardbasic bestehen in den Elementen flr strukturierte
Programmierung und darin, dass es keine Zeilennummern gibt. Ferner miissen — BASIC-untypisch —
Variablen im Voraus definiert werden. Dies hat seine Begriindung darin, dass der Compiler schlichtweg
einfacher ist, wenn man Variablen deklariert.

Bei den meisten mir bekannten BASIC-Dialekten konnte man kaum unterschiedliche Datentypen verwenden,
bestenfalls kann man zwischen Integer (= ganze Zahlen) und Float (= Gleitkommazahl) wahlen. Beim hier
vorliegenden BASIC hat man die Wahl zwischen Byte, Int, Long, Float und Double. Warum diese Vielfalt? Ich
mdchte eine moéglichst optimale Ausnutzung von Speicher und Prozessor erreichen. Wenn z.B. ein Array mit
1000 Elementen definiert wird, braucht es als long 4000 Bytes, als byte nur 1000 Bytes.



3.1 Datentypen:

Byte:

Ganze Zahlen mit einem Wertebereich von -128 bis +127. Bendtigt 1 Byte (8 Bit).

Int: (Integer)

Ganze Zahlen mit einem Wertebereich von -32768 bis +32767. Bendtigt 2 Byte (16 Bit).

Long: (long Integer)

Ganze Zahlen mit einem Wertebereich von -2147483648 bis +2147483647. Bendtigt 4 Byte (32 Bit).
Float: (Gleitkommazahl)

Gleitkommazahlen mit einem Wertebereich von (+/-)3,4*(+/-)10738. Bendtigt 4 Byte (32 Bit). Die Genauigkeit
liegt bei max. 7 Ziffern.

Double: (Gleitkommazahl)

Gleitkommazahlen mit einem Wertebereich von (+/-)1,7*(+/-)10*308. Bendtigt 8 Byte (64 Bit). Die
Genauigkeit liegt bei max. 15 Ziffern.

Vor der Programmierung sollte man sich im Klaren sein, welche Variablen welchen Zahlenbereich annehmen
kénnen und dementsprechend einen Datentyp wahlen. Die Benutzung von float und double sollte man so
weit als moéglich vermeiden, denn Gleitkommazahlen verbrauchen Gberdurchschnittlich viel Rechenleistung.

3.2 Rechenoperationen und Zahlen

Bei den Rechenoperationen ist (fast) alles erlaubt. Dieser Teil des Compilers hat mich einige
graue Haare gekostet, aber scheint nun zu funktionieren.
Es gilt: Punkt vor Strich, Klammern kénnen auch gesetzt werden.

+ Addieren

- Subtrahieren

* Multiplizieren

/ Dividieren

A Potenzieren

& Bitweise UND (nur fiir long, bei double ist das Ergebnis immer 0)
| Bitweise OR s.o.

! Bitweise XOR s.o.

&, | und ! funktionieren auch bei logischen Operationen (if, while etc.)

Funktionen:

sin() sinus

cos() cosinus

tan() tangens

cot() cotangens

asin() Umkehrfunktion zu sin()
acos() Umkehrfunktion zu cos()
atan() Umkehrfunktion zu tan()
acot() Umkehrfunktion zu cot()

Es gibt momentan einen bekannten Fehler im Parser:

Ist innerhalb einer Konstanten ein nicht fir Rechenoperationen reserviertes Zeichen (z.B. 4t5), wird kein
Fehler ausgegeben. Die Zahl wird bis zum Vorkommen des Zeichens ausgewertet, der Rest wird verworfen.
Im obigen Beispiel wirde die Zahl also als 4 interpretiert.

Zu beachten ist: Bei den Rechenoperationen mussen die unterschiedlichen Zahlentypen "gecastet", also
angepasst werden, sonst kann man long (Ganzzahl) und double (Gleitkomma) nicht miteinander verknupfen.
Dies macht der Interpreter automatisch. Dabei setzt sich immer das genauere (gréf3ere) Zahlenformat durch.
Die Reihenfolge: Byte — Integer — Long — Float — Double.

Kommt es bei dieser Anpassung zu Bereichsuberschreitungen, so zum Beispiel bei der Konvertierung von
Long in Byte, wird der jeweils maximale Wert iUbernommen, es wird keine Fehlermeldung ausgegeben.

Beispiel:
Long wird zu Byte
1374 127

-462 -128



Beispiele fir mathematische Ausdriicke:
a = (b+c)*(d+e)
a = 5*(b+8)*(-c-d)

Der Compiler nimmt dem Interpreter bereits die arbeit ab, die Zahlen aus dem Klartext in ein dem Rechner
verstandliches Format umzuwandeln. Konstanten werden direkt im Bytecode abgespeichert, wobei ganze
Zahlen 4 Bytes bendétigen, Gleitkomma 8 Bytes. Ist in einer Konstanten ein "." (Dezimalpunkt) vorhanden,
wird sie als double behandelt, sonst als long. Wenn man von vornherein weil3, dass eine Berechnung in
double ausgefiihrt wird, kann man die Laufzeit optimieren, indem man den Compiler dazu zwingt, eine double
abzuspeichern. Dann muss der Interpreter kein Typcasting mehr durchfuhren.

Es ist geplant, durch Bereichsprifung bei den Konstanten auch die fehlenden Datentypen byte, integer und

float zu implementieren.

1234  long - Konstante

1234.0 double

1234.0E4 double, entspricht 12340000

1234E4 long, das E wird nicht interpretiert, entspricht 1234

3.3 Strings

Strings (Texte) sind mit dem Robolnt-Basic auch mdéglich, allerdings mit der Einschréankung, dass sie nicht
dynamisch gespeichert werden, sondern einen festen Speicherbereich zugewiesen bekommen. Man muss
also die maximale Grofie des Strings angeben.

Ein String wird ahnlich behandelt, wie ein Array, deshalb geschieht auch die Definition des Strings mit dem
DIM — Befehl:

dim text (20) string
Dies definiert eine Variable mit dem Namen ,text* als String mit maximal 20 Zeichen Lange. Der String darf
wahrend des Programmablaufs auch kiirzer sein.

Mit Strings kann nur eine Rechenoperation ausgeflhrt werden, die Addition +.

Um Strings umzuwandeln, zu verandern stehen Funktionen zur Verfigung:

zahl = asc(zeichen) Wandelt ein Textzeichen in den entsprechenen Ascii-Code um.
zeichen = chr(zahl) Wandelt einen Ascii — Code in ein entsprechendes Textzeichen.
string = left(string1,anzahl) Ergibt einen Teilstring von string1 der Lange anzahl von links.
string = right(string1, anzahl) Ergibt einen Teilstring von string1 der Lange anzahl von rechts.
string = mid(string1,anfang,anzahl) Teilstring von string1, beginnend bei anfang, lange = anzahl.
zahl = val(string) Wandelt einen String in eine Zahl

zahl = valf(string) wie oben, fir Gleitkommazahlen

string = str(zahl) Wandelt eine Zahl in einen String

string = strf(zahl) wie oben, fur Gleitkommazahlen

zahl = len(string) Gibt die tatsachliche Lange des Strings zurlick

Um Strings zuzuweisen, wird der Text innerhalb Anfiihrungszeichen angegeben.

Text = “Dies ist ein Test®

(Gestrichen: Wird die vordefinierte Lange eines Strings Uberschritten, wird ein Runtime-Fehler ausgeldst.)
Der Stringstack ist 127 Zeichen grof3, d.h. Additionen von Strings dirfen nicht Uber eine Lange von 127
Zeichen gehen.

Es wird bei Uberschreitung kein Runtime-Fehler ausgelst. Zu lange Strings werden einfach
abgeschnitten. Dies gilt auch bei Uberschreitung der Léange von Stringvariablen.

3.4 Reihenfolge der Definitionen

Der Compiler akzeptiert die verschiedenen Definitionen nur in einer bestimmten Reihenfolge.



. Globale Variablen

Pob=

Hauptprogramm (main:)

Funktionen (mit eigenen globale Variablen)
. Tasks (mit eigenen globale Variablen)

Grundsatzlich muss eine Funktion definiert sein, bevor sie aufgerufen werden kann. Ein rekursiver Aufruf (die
Funktion ruft sich selbst auf) ist deshalb mdglich, da die Definition vor dem eigentlichen Aufruf innerhalb der

Funktion steht.

Die Definitionen von Tasks und Funktionen kénnen miteinander vertauscht werden, sollen Funktionen aus
Tasks heraus aufgerufen werden, missen die Funktionen wiederum vor dem Task definiert sein.

Globale Variablen missen direkt nach der Kopfzeile der Funktion / des Tasks definiert werden. Hierzu gibt es

im Anhang noch einige Beispiele.

Derzeit muss der gesamte Quellcode in einer einzigen Datei vorliegen, eine Aufsplittung ist geplant,
allerdings habe ich noch keine Lésung, wie in diesem Fall Zeilennummern bei Fehlern behandelt werden

konnten.

4 Befehlsreferenz

4.1 Ubersicht (nach Funktion sortiert)

4.1.1 Ein- und Ausgabe

motor nummer, wert
out nummer, wert

wert = in (nummer)

wert = analog (nummer)
wert = ir

print Ausdruck

put wert

wert = get()

string = input()

baud wert

4.1.2 Programmsteuerung

if (Bedingung) ... else ... end
repeat .... until (Bedingung)
while (Bedingung) end
select .... case ... end

edge nummer, typ, time-out
for ... next

function (....)

waitms zeit

4.1.3 Variablendefinition

var name typ
dim name (Anzanhl) typ
dim name (Anzahl1,Anzahl2) typ

4.1.4 Multitasking

task name
start name

Motor (nummer) wird mit Stufe (wert) angesteuert

Ausgang (nummer) wird mit Stufe (wert) angesteuert

Abfrage des Eingangs (nummer), Riickgabe 0 oder 1

Abfrage des analogen Eingangs (nummer)

Abfrage des Infrarot-Eingangs

Ausgabe auf die serielle Schnittstelle des Robolnt

Ausgabe auf die serielle Schnittstelle des Robolnt (ein Zeichen)
Lesen eines Zeichens von der seriellen Schnittstelle

Lesen einer Zeile von der seriellen Schnittstelle

Einstellen der Baudrate der seriellen Schnittstelle

Verzweigung
Schleife
Schleife
Mehrfache Verzweigung
wartet auf Ereignis an einem Eingang
Schleife
Definiert eine Funktion (Unterprogramm)
Wartet die Zeit (angegeben in ms) ab.

Definiert einen Variable
Definiert ein eindimensionales Feld, Sonderfall: String.
Definiert ein zweidimensionales Feld.

Definiert einen Task
Startet den Task ,name”



stop name Stoppt den Task ,name* (Pause)
kill name Beendet den Task ,name*

4.1.5 Mathematische Funktionen

ergebnis = sin(wert) Berechnet den Sinus.

ergebnis = cos(wert) Berechnet den Cosinus.
ergebnis = tan(wert) Berechnet den Tangens.
ergebnis = cot(wert) Berechnet den Cotangens.
ergebnis = asin(wert) Umkehrfunktion fir Sinus.
ergebnis = acos(wert) Umkehrfunktion fiir Cosinus.
ergebnis = atan(wert) Umkehrfunktion fiir Tangens.
ergebnis = acot(wert) Umkehrfunktion fiir Cotangens.

4.1.6 Stringfunktionen

string = left(text, anzahl) Gibt (anzahl) Zeichen von links an zurlick
string = right(text, anzahl) Gibt (anzahl) Zeichen von rechts an zuriick
string = mid(text, start, anzahl) Gibt (anzahl) Zeichen von (start) an zurlick
zahl = val(string) Wandelt einen String in eine Zahl (Ganzzahl)
zahl = valf(string) Wandelt einen String in eine Zahl (Gleitkomma)
string = str(zahl) Wandelt eine Zahl in einen String (Ganzzahl)
string = strf(zahl) Wandelt eine Zahl in einen String (Gleitkomma)
string = chr(zahl) Wandelt eine Zahl in einen Buchstaben

zahl = asc(string) Wandelt einen Buchstaben in eine Zahl

4.2 Ausfuhrliche Referenz (alphabetisch)

<Ausdruck>

Ein Ausdruck kann aus Variablen, Konstanten oder Kombinationen davon bestehen (z.B. eine komplette
Berechnungsformel). Der jeweilige Befehl verwendet dann das Ergebnis der Berechnung als Parameter.
Ebenfalls erlaubt sind Vergleiche, logische Verknlpfungen.

Gliltige Ausdriicke:
10 Konstante
a Variable.
ath Formel
(a+35)*(b+2)

a=0

b>5

(atb)=(c " d)
in(x)=0

sin(0) > 1

b and 2

Bei Bedingungen z.B. b>5 ist der Wert des Ausdrucks 1, wenn die Bedingung erfullt ist, ansonsten 0.
Zuweisungen wie

a=b>5

sind erlaubt, a wird eine 1 zugewiesen, wenn b > 5 erfilllt ist, ansonsten wird a eine 0 zugewiesen.

<variable>

Ein Variablenname kann aus bis zu 15 Zeichen bestehen. Der Variablenname muss mit einem Buchstaben
beginnen und darf auch Sonderzeichen (z.B. Umlaute) enthalten. Die Grof3- / Kleinschreibung wird bei den
Variablennamen nicht berlicksichtigt, die Variable ,Zahler” und ,zahler* werden nicht unterschieden.



<Ergebnis> = acos(<Ausdruck>)

Berechnet den Arcuscosinus von <Ausdruck>. Umkehrfunktion von cos().
Das Ergebnis ist vom Typ double.

<Ergebnis> = acot(<Ausdruck>)

Berechnet den Arcuscotangens von <Ausdruck>. Umkehrfunktion von cot().
Das Ergebnis ist vom Typ double.

<Ergebnis> = analog(<Ausdruck>)
Fragt einen der analogen Eingange des Robo Interface ab. <Ausdruck> gibt die Nummer des

Analogeingangs an. Um die Programmierung zu vereinfachen, sind Konstanten vordefiniert. Eine Liste der
Konstanten finden man im Anhang.

<Ergebnis> = asin(<Ausdruck>)

Berechnet den Arcussinus von <Ausdruck>. Umkehrfunktion von sin().
Das Ergebnis ist vom Typ double.

<Ergebnis> = atan(<Ausdruck>)

Berechnet den Arcustangens von <Ausdruck>. Umkehrfunktion von tan().
Das Ergebnis ist vom Typ double.

baud <Ausdruck>

Setzt die Baudrate der seriellen Schnittstelle auf den Wert <Ausdruck>.
Es sind nahezu beliebige Werte einstellbar, sinnvoll sind Standardwerte:
1200, 2400, 4800, 9600, 19200, 38400

<Ergebnis> = cos(<Ausdruck>)

Berechnet den Cosinus von <Ausdruck>.
Das Ergebnis ist vom Typ double.

<Ergebnis> = cot(<Ausdruck>)

Berechnet den Cotangens von <Ausdruck>.
Das Ergebnis ist vom Typ double.

dim <variable> (<Konstante>[,<Konstante>]) <typ>

Definiert ein Array (Feld). Felder sind immer global und miissen an der entsprechenden Stelle im Programm
(direkt am Anfang) definiert werden. Arrays kdnnen ein- oder zweidimensional sein.

Gerade bei Arrays ist der Speicherbedarf sehr grof3, deshalb sollte der Typ des Arrays sorgfaltig gewahlt
werden.

Beispiele:

dim positionen (100,4) int
dim werte (20) float

Sonderfall: Strings



dim text (20) string

Strings durfen nur eine Dimension haben.

edge <Ausdruck1>, <Ausdruck2>[, <Ausdruck3>]

Ausdruck1 = Nummer des abzufragenden Digital-Eingangs
Ausdruck2 = Typ (-1,0 oder 1, vordefinierte Konstanten #down, #any, #up)
Ausdruck3 = Timeout in ms (optional)

Wartet auf eine Flanke am angegeben Eingang. Als Typ kann #any (irgendeine), #up (steigend, also 0->1)
oder #down (fallend, also 1->0) angegeben werden.

Zusatzlich kann mit Timeout die maximale Wartezeit in Millisekunden auf die Flanke angegeben werden.
Dies ist nutzlich, um zum Beispiel einen blockierten Motor zu erkennen.

Der maximale Wert fur Timeout ist 65535.

for <variable> = <Ausdruck 1> to <Ausdruck 2> [step <Ausdruck 3>]
<Anweisungsblock>
next

Fihrt einen Anweisungsblock ein oder mehrfach aus, bis die Endbedingung <variable> = <Ausdruck 2>
erreicht ist. ,step” gibt die Schrittweite an, wird ,step“ weggelassen, wird +1 angenommen.

For-Next Schleife. Die Variable <variable> wird von <Ausdruck 1> bis <Ausdruck 2> mit der Schrittweite
<Ausdruck 3> durchgezahilt.

Bei negativen <step> kann auch ruckwarts gezahlt werden. Die Angabe von <step> ist optional,

wird sie weggelassen, wird automatisch 1 angenommen. Hinter next wird nicht, wie teilweise Ublich, die
Zahlvariable angegeben - sie wird automatisch zugewiesen.

(Einschalten der Motoren 1 bis 4)

forx=1to 4
motor x,8
next

(Zahlen von 10 bis 0 riickwarts)

for x=10 to O step -1
<Anweisungsblock>
next

(Verschachtelte Schleifen)

for x=1 to 10 step 2
for y=20 to 25
<Anweisungsblock>
next
next

(Schrittweiten kleiner als 1)
forx=1.0 to 1.5 step 0.1
<Anweisungsblock>

next

Wichtig: Der Anweisungsblock wird auf jeden Fall mindestens einmal ausgefihrt, auch wenn die
Schleifenbedingung nicht erfullt ist.

function <Funktionsname> [(<variable> <typ>, ...) [typ]]



Definiert eine Funktion mit dem entsprechenden Namen.
Eine Funktion kann keine oder bis zu vier Ubergabeparameter haben und / oder bis zu einen
Rickgabeparameter.

Beispiele:

function test Definiert eine Funktion ohne Parameter

function test1 (a long,b float) Definiert eine Funktion mit zwei Ubergabeparametern
function test2 (a long) long Definiert eine Funktion mit einem Ubergabe- und einem

Ruckgabeparameter
Beispiel fur eine Funktion mit Parametern:

function summe (a long, b long) long
return a+b

Der Aufruf dieser Funktion:
x = summe (10,20)
Beispiel fir eine rekursive Funktion (gesamtes Programm)

var x long
function summe (parameter long) long
var a long
var b long
if parameter <=1
a=1
else
b = summe (parameter — 1)
a = parameter + b
end
return a

main:
print summe (10)

Tipp: Es fallt auf, dass im else — Teil des Unterprogramms unnétigerweise die Variable b benutzt wird.
Kirzer kdnnte es heil3en:

else
a = parameter + summe(parameter — 1)
end

Dies hat allerdings den Nachteil, dass hierdurch ein Taskwechsel verhindert wird, weil die Variable parameter
zur Berechnung auf den mathematischen Stack gelegt wird. Erst nach Rlckkehr der Funktion (also nach
Ausfihrung aller Rekursionen) ist der mathematische Stack bereinigt. AuRerdem hat der mathematische
Stack nur 10 Speicherplatze, eine Berechnung von summe (11) ware nicht mdglich. Durch das Speichern
des Ergebnisses in der lokalen Variable b werden diese Effekte verhindert.

Diese Eigenart kommt bei normaler Nutzung kaum zu tragen, sollte es zu Uberdurchschnittlich vielen
Problemen fiihren, kann sie durch eine Anderung am Interpreter eliminiert werden.

<Ergebnis> = get()

Liest ein einzelnes Zeichen von der seriellen Schnittstelle. Es wird nicht auf ein Zeichen gewartet, sondern
die Funktion gibt den Wert -1 zurlck, falls kein Zeichen von der seriellen Schnittstelle empfangen wurde.
Das Ergebnis ist vom Typ int.

If <Vergleich>
<Anweisungsblock1>
[else



<Anweisungsblock2>]
end

JFalls ... dann ... sonst...”

Falls die im Vergleich angegebene Gleichung wahr ist, wird Anweisungsblock 1 ausgefuhrt, ansonsten
(sofern vorhanden) Anweisungsblock 2.

Ein Anweisungsblock kann aus einem oder mehreren Anweisungen bestehen. Es durfen beliebig viele if-
else-end Anweisungen ineinander verschachtelt werden.

Beispiel A:
ifa=0

motor 1,0 // aist gleich O
else

motor 1,8 // aist ungleich 0
end

Falls a = 0, wird der Motor abgeschaltet, bei allen anderen Werten von a wird er eingeschaltet.

Beispiel B:
ifa=0

motor 1,0
end

Falls a = 0, wird der Motor abgeschaltet, bei allen anderen Werten von a passiert nichts (es wird der nachste
Befehl nach end ausgefihrt).

<Ergebnis> = in (<Ausdruck>)

Fragt die digitalen Eingange des Robo Interface ab. <Ausdruck> kann Werte zwischen 1 und 32 annehmen,
das Ergebnis ist entweder 0 (Eingang nicht betatigt) oder 1 (Eingang betatigt).

Beispiele:

a=in(8) Fragt Eingang 8 ab und speichert das Ergebnis in a

ifin(1)=1 Schaltet den Motor 1 ein, wenn Eingang 1 betatigt wird.
motor 1,8

end

<Ergebnis> = input()

Liest einen String von der seriellen Schnittstelle. Der Befehl wartet so lange, bis ein Zeilenvorschub
empfangen wird.

<Ergebnis> = ir

Fragt dgn Infrarot-Eingang des Robo Interface ab. Ist keine Taste auf der Fernbedienung gedriickt, ist das
E:geﬁ?g;rsag.en zu vereinfachen, sind Konstanten vordefiniert. Eine Liste der Konstanten findet man im

Anhang.

kill <taskname>

Beendet einen Task und initialisiert ihn neu. Bei einem weiteren start <taskname> startet der Task
neu.

motor <Ausdruck 1>,<Ausdruck 2>



Steuert einen Motorausgang des Robo Interface an. <Ausdruck 1> gibt die Nummer des Motors an (1..16),
<Ausdruck 2> die Drehrichtung und Geschwindigkeit. <Ausdruck 2> kann Werte zwischen -8 und 8
annehmen, bei negativen Werten dreht der Motor links herum, bei positiven rechts herum. Bei O wird der
Motor abgeschaltet. GrolRere oder kleinere Werte bei den Parametern werden ignoriert.

Beispiele:
motor 1,8 Schaltet Motor 1 ein, rechtsdrehend, mit maximaler Geschwindigkeit.
motor a,b Je nach Inhalt der Variablen a und b wird ein Motor ein- oder ausgeschaltet.

out <Ausdruck 1>,<Ausdruck 2>

Steuert einen Ausgange des Robo Interface an. <Ausdruck 1> gibt die Nummer des Ausgangs an (1..32),
<Ausdruck 2> die Geschwindigkeit. <Ausdruck 2> kann Werte zwischen 0 und 8 annehmen. Bei 0 wird der
Ausgang abgeschaltet.

print <Ausdruck>

Zwar hat das Robo Interface keinen Bildschirm, aber eine serielle Schnittstelle. Mit diesem Befehl kdnnen
Variablenwerte oder Konstanten (z.B. zur Fehlersuche) Uiber die serielle Schnittstelle ausgegeben werden.
Die serielle Schnittstelle ist mit 38400 Baud, No Parity, 1 Stopbit vordefiniert.

Beispiele:

print a Gibt den Inhalt der Variablen a aus.

print a+5 Addiert 5 zur Variablen a hinzu und gibt das Ergebnis aus.
print 10 Gibt die Konstante 10 aus.

print “Hallo” Gibt den Text ,Hallo* aus.

Nach jedem Print-Befehl wird ein Zeilenvorschub gesendet, so dass die nachste Ausgabe in eine neue Zeile
geschrieben wird. Mit dem Semikolon am Ende des auszugebenden Textes kann diese Ausgabe unterdriickt
werden.
Beispiel:

print “Hal*;
print “lo* Gibt den Text ,Hallo® aus (kein Zeilenvorschub zwischen den Print-Befehlen)

put <Ausdruck>

Gibt ein einzelnes Zeichen auf der seriellen Schnittstelle aus. <Ausdruck> muss ein Zahlenwert sein.

repeat
<Anweisungsblock>
until <Ausdruck>

Fihrt den Anweisungsblock so lange aus, bis der Ausdruck wahr (ungleich 0) ist.
Beispiel:
a=1 Stoppt die Motoren 1 - 4
repeat
motor a,0
a=a+1
until a >=4

Der Anweisungsblock wird mindestens ein Mal ausgefiihrt.

Select <Ausdruck>



case <Konstante 1>
<Anweisungsblock 1>

case <Konstante 2>
<Anweisungsblock 2>

end
Mit der ,select — case” - Anweisung kénnen Programmverzweigungen elegant geldst werden. Beispiel: Die
Abfrage der IR-Schnittstelle. Hierbei wird der Ausdruck mit den Konstanten verglichen und bei Gleichheit der

entsprechende Anweisungsblock ausgefihrt. Gleich danach springt die Programmausfiihrung komplett an
das Ende der Select-Anweisung, es werden also keine weiteren Vergleiche durchgefuhrt.

select ir Abfrage des IR - Eingangs
case #m1l1 ist die Taste ,Motor 1 Links" gedrlickt ?
motor 1,-8 Motor 1 linksherum einschalten
while ir <> 0 warten, bis die Taste wieder losgelassen wird
end
motor 1,0 Motor 1 abschalten
case #m1ir1 das Gleiche flr rechtsherum
motor 1,8
while ir <> 0
end
motor 1,0
case #m2l1 weitere Abfragen kdnnen folgen
-case #m2r1
énd

<Ergebnis> = sin(<Ausdruck>)
Berechnet den Sinus von <Ausdruck>.
Das Ergebnis ist vom Typ double.
start <taskname>

Startet einen Task.Wurde der Task vorher durch ,stop“ unterbrochen, wird er an der gleichen Stelle
ausgefuhrt, an der er unterbrochen wurde. Ein Task, dessen Ausfihrung beendet ist (keine Schleife), muss
erst durch kill* neu initialisiert werden.

stop <taskname>

Stoppt einen Task. Wird der task mit start <taskname> gestartet, macht er dort weiter, wo er aufgehért
hat. stop <taskname> ist so etwas wie eine "pause" - Taste.

<Ergebnis> = tan(<Ausdruck>)

Berechnet den Tangens von <Ausdruck>.
Das Ergebnis ist vom Typ double.

task <taskname>



Definiert einen Task. Man kann sich einen Task vorstellen, wie ein Unterprogramm, das zwischendurch
immer wieder aufgerufen wird. Die Definition muss immer durch ein "end" beendet werden.

Siehe auch : start, stop, kill

Wird der Programmiteil innerhalb des Tasks nicht durch eine Schleife wiederholt (z.B. durch ein goto), wird
der Task nur ein einziges mal ausgefihrt, kann aber durch ,kill“ und anschlieRendes ,start* neu gestartet
werden.

Innerhalb eines Tasks kénnen lokale Variablen definiert werden. Wichtig ist: Die Variablendefinition muss
direkt hinter der Taskdefinition stehen.

Beispiele:
Task ohne lokale Variablen, nur einmal ausgefiihrt:

task reset
motor 1,0
end

Sobald bei der Ausflihrung das ,end” erreicht wird, bleibt der Task an dieser Stelle stehen.
Task ohne lokale Variablen, dauernd ausgefuhrt:

task blinken
while 1
output 1,8
waitms 200
output 1,0
waitms 200
end
end

Task mit lokalen Variablen:

task test
var a long
var b long
a=1
b=0
while 1
motor a,b
a=a+1
ifa>4
a=1
end
end
end

var <variable> <typ>

Definiert (je nach Position innerhalb des Programmes) eine globale oder lokale Variable.
Beispiele:

var variable1 long
var x int
var y double

Variablennamen missen immer mit einem Buchstaben beginnen und diirfen anschlieRend nur Buchstaben,
Zahlen oder den Unterstrich _ beinhalten.

Direkt am Programmbeginn (also vor allen Funktionen und Tasks) definierte Variablen sind global fur das
gesamte Programm, d.h. sie sind Gberall im Programm guiltig.

waitms <Ausdruck>



Wartet die angegebene Zeit in ms. Wahrenddessen werden andere Tasks weiter ausgefiihrt.
Beispiele:

waitms 1000
waitms a

while <Ausdruck>
<Anweisungsblock>
end

Fuhrt den Anweisungsblock so lange aus, wie der Vergleich wahr ist. Ist der Vergleich beim ersten Durchlauf
bereit unwahr, wird der Anweisungsblock Uberhaupt nicht ausgefuhrt (im Gegensatz zu repeat — until).

Endlosschleifen (z.B. fur tasks) kdnnen mit der while-Schleife sehr einfach realisiert werden:

while 1
<Anweisungsblock>
end

Die 1 bedeutet in diesem Fall, dass der Vergleich immer wahr ist, also wird diese Schleife niemals verlassen.

Momentan ist der zur Unterbrechung von while-Schleifen Ubliche break-Befehl nicht implementiert, aber Uber
eine zusatzliche Variable kann diese Funktionalitat nachgebildet werden:

breakvar =1
while breakvar
<Anweisungen>
if <Abbruchbedingung>
breakvar =0
end
end

5. Beispielprogramme

Dieses Programm schaltet Motor 1 und 2 links oder rechts abhangig
vom Zustand der Eingdnge 1 und 2. Wird Eingang 8 betatigt, wird das
Programm beendet.

main:
if in(1)=1
motor 1,8
else
motor 1,-8
end
if in(2)=1
motor 1,8
else
motor 1,-8
end
if in(8)=0
goto main
end

Das gleiche Programm, nur mit Multitasking mit einem Task fur jeden
Motor.

task motor1



while 1
if in(1)=1
motor 1,8
else
motor 1,-8
end
end
end

task motor2
while 1
if in(2)=1
motor 2,8
else
motor 2,-8
end
end
end

main:

start motor1
start motor2
while in(8) =0
end

Fir dieses Programm sollte man 8 Lampen an die Ausgange anschliel3en.
Die Lampen blinken unabhangig voneinander durch die unterschiedlichen
Wartezeiten. Dies ist der Funktionstest fir die Multitasking-Funktion.
task motor1
while 1
motor 1,8
waitms 500
motor 1,-8
waitms 500
end
end

task motor2
while 1
motor 2,8
waitms 525
motor 2,-8
waitms 525
end
end

task motor3
while 1
motor 3,8
waitms 550
motor 3,-8
waitms 550
end
end

task motor4
while 1
motor 4,8
waitms 575



motor 4,-8
waitms 575
end
end

main:

start motor1
start motor2
start motor3
start motor4
while 1

end

Weitere Beispiele befinden sich im Dateiordner ,Testprogramme*

ANHANG

Vordefinierte Konstanten

Konstanten beginnen immer mit #

#ax (
#ay (
#az (
#a1 (
#a2 (
#av (
#d1 (
#d2  (
#axs1 (
#axs2 (
#axs3 (
#avs1 (
#avs2 (
#avs3 (

~— — ~— ~—

#m3r1 (1)
#m3l1 (2)
#m1spd1 (3)
#m2spd1 (4)
#m3spd1 (5)
#code2 (6)
#m1r1

#up (1)
#any (0)
#down (-1)

Analog-Eingang AX am Robo-Interface
Analog-Eingang AY am Robo-Interface
Analog-Eingang AZ am Robo-Interface
Analog-Eingang A1 am Robo-Interface
Analog-Eingang A2 am Robo-Interface
Spannungsversorgung des Robo-Interface
Distanzsensor-Eingang 1

Distanzsensor-Eingang 2

Analog-Eingang AX am ersten Extension-Modul
Analog-Eingang AX am zweiten Extension-Modul
Analog-Eingang AX am dritten Extension-Modul
Spannungsversorgung am ersten Extension-Modul
Spannungsversorgung am zweiten Extension-Modul
Spannungsversorgung am dritten Extension-Modul

IR-Fernbedienung Taste Motor 3 rechts
IR-Fernbedienung Taste Motor 3 links
IR-Fernbedienung Taste Geschwindigkeit Motor 1
IR-Fernbedienung Taste Geschwindigkeit Motor 2
IR-Fernbedienung Taste Geschwindigkeit Motor 3
IR-Fernbedienung Taste Code2
IR-Fernbedienung Taste Motor 1 rechts
IR-Fernbedienung Taste Motor 1 links
IR-Fernbedienung Taste Motor 2 rechts
IR-Fernbedienung Taste Motor 2 links
IR-Fernbedienung Taste Code1

Fur Edge-Funktion



